

Excel avancé 3

Ahmed Laatabi

ENSAM – Meknès

2024 - 2025

Objectifs

- Maîtriser les notions avancées de Microsoft Excel
- Organiser des données et analyser une problématique professionnelle pour la transposer systématiquement sous Excel.
- Créer des indicateurs clés, analyser des données, bien présenter ses résultats.

Programme

1. Rappel des fonctions de base d'Excel (références absolues, moyennes, médianes, sommes conditionnelles, SI, ...)
2. Filtres et tris avancés, mises en forme conditionnelles, styles personnalisés.
3. Formules complexes et multicritères, imbrications de SI, autres imbrications.
4. Formules conditionnelles, recherche H et V.
5. Formules matricielles, UNIQUE, TROUVE, ...
6. Menus multi-déroulants de saisie, verrouillage de cellules, organisation des données, protection des feuilles classeurs.
7. Utilisation des dates, années, jours, mois, heures, calculs avec ces formules imbriquées.
8. Graphiques évolués, superposition de graphes, graphes multi-échelles, choix et interprétation de graphiques, analyse, mise en forme de données.
9. Tableaux croisés dynamiques (TCD), graphiques croisés dynamiques (GCD).
10. Macros enregistrés.
11. Initiation aux macros programmés en VBA.
12. Raporting, création d'indicateurs clés, tableaux de bords, méthodologie, analyse de données, aide à la décision, études de cas.

Conditions et références

=COUNTIF(A2:A151,">5")

=COUNTIF(A2:A151,">"&A2)

=COUNTIF(A3:A152,">"&A3)

=COUNTIF(A2:A151,">"&\$A\$2)

Arguments de fonctions

- Dans la signature d'une fonction, les arguments *facultatifs* sont entre crochets [].
 - La fonction fonctionne sans ces arguments.
 - S'ils sont omis, Excel prend les valeurs par défaut.
- On peut se passer d'un argument facultatif en laissant sa place vide.

INDEX(array, row_num, [column_num])

- Les trois écritures sont valides, et renvoient le même résultat car la plage passée à la fonction est sur une seule ligne.

=**INDEX**(A2:D2,3)

=**INDEX**(A2:D2,**1**,3)

=**INDEX**(A2:D2,,3)

PEMDAS et Modulo

Dans la majorité des langages de programmation, *Modulo* (%) est situé dans le même rang que la multiplication et la division.

- Dans Excel, l'opérateur (%) est utilisé pour les pourcentages :

$$= 2 \% // \rightarrow 0.02$$

- La fonction **MOD** permet de trouver le modulo :

$$\text{MOD}(\text{number}, \text{divisor})$$

$$= 3 + \text{MOD}(3, 2) // \rightarrow 4$$

Programme

1. Rappel des fonctions de base d'Excel (références absolues, moyennes, médianes, sommes conditionnelles, SI, ...)
2. Filtres et tris avancés, mises en forme conditionnelles, styles personnalisés.
3. Formules complexes et multicritères, imbrications de SI, autres imbrications.
4. Formules conditionnelles, recherche H et V.
- 5. Formules matricielles, UNIQUE, TROUVE, ...**
6. Menus multi-déroulants de saisie, verrouillage de cellules, organisation des données, protection des feuilles classeurs.
7. Utilisation des dates, années, jours, mois, heures, calculs avec ces formules imbriquées.
8. Graphiques évolués, superposition de graphes, graphes multi-échelles, choix et interprétation de graphiques, analyse, mise en forme de données.
9. Tableaux croisés dynamiques (TCD), graphiques croisés dynamiques (GCD).
10. Macros enregistrés.
11. Initiation aux macros programmés en VBA.
12. Raporting, création d'indicateurs clés, tableaux de bords, méthodologie, analyse de données, aide à la décision, études de cas.

Opérations sur les matrices

- **MMULT** : multiplier deux matrices.
 - Dans le choix des plages, il **faut respecter les conditions du calcul matriciel**, si non la formule renvoie une erreur. Il faut également sélectionner une plage appropriée pour recevoir l'ensemble du résultat.
// multiplie une 3x2 par une 2x4 donc elle donne une 3x4
{=MMULT(A3:B5,B8:D9)}
- **TRANSPOSE** : transposer une matrice. *// une 3x2 donne une 2x3*
 - Fonctionne aussi sur des cellules avec du **texte** (utile pour **V** → **H**) !
- **MINVERSE** : inverser une matrice. *// une **2x2** donne une 2x2*
- **MDETERM** : calculer le déterminant d'une matrice. *// valeur unique*

Formules matricielles

- Les *formules matricielles* permettent d'effectuer des calculs sur plusieurs valeurs/lignes en même temps, plutôt que de calculer une valeur/ligne unique.
- Utiles pour les opérations sur les ensembles de données, notamment des *vecteurs (tableaux)* ou des *matrices*.
- *Formule matricielle* (**array formula**) : traite des tableaux, pas uniquement des matrices.
- Dans les anciennes versions d'Excel, une formule matricielle doit être validée avec **Ctrl + Shift + Enter**. Après validation, Excel l'encadre automatiquement de crochets {}.
- **Faire attention** aux formules qui renvoient un résultat sur une plage au lieu d'une seule cellule.

Formule normale vs matricielle

- Formule normale : renvoie le résultat sur une seule cellule :

=**SUM**({ 1,2,3 } * 4) → *une seule cellule : 24 = 4+8+12*

- Formule normale : renvoie le résultat sur plusieurs cellules :

{= { 1,2,3 } * 4 } // → *une matrice 1x3 : {4,8,12}*

On ne peut pas modifier le contenu de cellules résultats d'une formule matricielle :

"You cannot change part of an array".

→ *Suppression complète de la matrice.*

Formules matricielles

- La somme des produits de *sepal_length* (A2:A151) x *sepal_width* (B2:B151) :
 - Fonction **SOMMEPRODUIT**, formule normale :
=SUMPRODUCT(A2:A151,B2:B151)
 - Fonction **SOMME**, formule matricielle : chaque élément de la plage A2:A151 est multiplié par l'élément correspondant dans la plage B2:B151 (même ligne), puis la somme des produits est calculée. Attention, si la formule n'est pas validée *matriciellement* **{}**, elle donne comme résultat le produit des deux cellules qui correspondent à la cellule de la formule.

{=SUM(A2:A151*B2:B151)}

// plus lisible avec les parenthèses

{=SUM((A2:A151)*(B2:B151))}

Formules matricielles

- Les formules matricielles peuvent être unidimensionnelles ou bidimensionnelles.
- **Unidimensionnelles** : prennent comme arguments des *vecteurs*, c-à-d. des plages d'une même colonne ou d'une même ligne.
 - La somme des produits des colonnes A et B pour les *Iris-setosa* uniquement :
$$\{=SUM((A2:A151)*(B2:B151)*(E2:E151="Iris-setosa"))\}$$
- **Bidimensionnelles** : permettent d'effectuer des calculs sur des matrices complètes (plages sur plusieurs lignes ou colonnes).
 - La somme des produits de plages multi-colonnes et multi-lignes. SOMMEPRODUIT ne fait pas une multiplication matricielle, mais multiplie les éléments correspondants (dans les deux plages) et fait la somme. Les deux plages doivent avoir les mêmes dimensions.

$$=SUMPRODUCT(A2:B3,C2:D3) // =A2*C2+B2*D2+A3*C3+B3*D3$$

En pratique

- Les formules matricielles permettent de faire des calculs complexes sans avoir besoin de créer des *colonnes intermédiaires*.

Maximum de différence entre A et B sans passer par une colonne intermédiaire

`{=MAX((A2:A151) - (B2:B151))}`

- **SOMMEPRODUIT** avec des conditions...
- La somme des carrées d'une plage de cellules :

`{=SUM(A2:A151^2)}` *ou* **`{=SUM(A2:A151*A2:A151)}`**

- L'écart type de *sepal_length* :

`{=SQRT(SUM((A2:A151-AVERAGE(A2:A151))^2)/150)}`

ou **`=STDEV.P(A2:A151)`** // *=STDEV.S(A2:A151) ?*

FREQUENCY

- **FREQUENCY** est une fonction matricielle calcule la **distribution de fréquence** d'un ensemble de données en fonction d'intervalles donnés (renvoie le nombre de valeurs qui appartient à chaque intervalle).

FREQUENCY(data_array, bins_array)

- *data_array* : une plage de colonnes contenant les données (ex : A2:A151).
 - *bins_array* : la plage contenant les bornes (limites) des intervalles, triées *obligatoirement en ordre croissant*. **Le nombre d'intervalles est le nombre de bornes + 1.**
- Comme toutes les formules matricielles dans Excel 2016, il faut sélectionner une plage appropriée pour pouvoir l'ensemble du résultat. À *partir de Excel 365, les formules se propagent automatiquement*.

FREQUENCE

- *Compter le nombre de fleurs dont le `sepal_length` appartient aux intervalles limités par 3,5 et 7 ?*
 - Remplir la plage **I2:I4** par les limites : 3,5, 7.
 - Sélectionner une plage de 4 cellules dans le même mode que la plage des limites (ici, *une même colonne* à l'instar de **I2:I4**).

{=FREQUENCY(A2:A151,I2:I4)}

// **0** : nombre de `sepal_length` <=3

// **32** : nombre de `sepal_length` >3 et <=5

// **106** : nombre de `sepal_length` >5 et <=7

// **12**: nombre de `sepal_length` >7

*Vérifier le résultat par des **COUNTIF** et **COUNTIFS** ..*

UNIQUE

- **UNIQUE** : renvoie une liste d'éléments uniques à partir d'une plage de cellules. Peut être utilisée, par exemple, pour nettoyer les doublons. *Elle est disponible à partir d'Excel 365 et Excel 2021.*

UNIQUE(plage, [par_colonne], [exactement_une_fois])

- *par_colonne*: extraire les valeurs uniques par ligne (FALSE, *par défaut*) ou par colonne (TRUE).
- *exactement_une_fois* : si FALSE (par défaut), elle extrait des valeurs uniques de la plage (annule les duplicates). Si TRUE, elle ne renvoie que les éléments qui apparaissent exactement une seule fois dans la plage.

TROUVE et CHERCHE

- **TROUVE** (*FIND*) : cherche un texte dans une chaîne de texte plus grande.
 - Elle renvoie la position de la première occurrence du texte recherché (indice du premier caractère), en commençant la recherche à partir du caractère *start_num* (0 par défaut).
 - Elle est sensible à *la casse*.

FIND(find_text, within_text, [start_num])

- **CHERCHE** (*SEARCH*) a le même fonctionnement (et la même syntaxe) que TROUVE, mais elle n'est pas sensible à la casse.

=**FIND**("setosa",E2) // → 6

=**FIND**("Setosa",E2) // #VALUE!

=**SEARCH**("Setosa",E2) // → 6

Imbrications ...

- FIND renvoie un nombre (la position) qu'on peut convertir en un booléen qui est à TRUE si la chaîne contient le texte recherché :

=**ISNUMBER**(**FIND**("setosa",E2)) // → *TRUE*

- Ou voir si la formule génère une erreur :

=**IFERROR**(**FIND**("Setosa",E2),"Texte introuvable") // #N/A → *Texte introuvable*

- La fonction **SIERREUR** (*IFERROR*) renvoie une valeur spécifique si une formule génère une erreur. Sinon, elle renvoie le résultat normal de la formule.

=**IFERROR**(formule, valeur_si_erreur)

=**IFERROR**(4/0, "Erreur de division") // → *Erreur de division*

=**IFERROR**(4/2, "Erreur de division") // → 2

Programme

1. Rappel des fonctions de base d'Excel (références absolues, moyennes, médianes, sommes conditionnelles, SI, ...)
2. Filtres et tris avancés, mises en forme conditionnelles, styles personnalisés.
3. Formules complexes et multicritères, imbrications de SI, autres imbrications.
4. Formules conditionnelles, recherche H et V.
5. Formules matricielles, UNIQUE, TROUVE, ...
- 6. Menus multi-déroulants de saisie, verrouillage de cellules, organisation des données, protection des feuilles classeurs.**
7. Utilisation des dates, années, jours, mois, heures, calculs avec ces formules imbriquées.
8. Graphiques évolués, superposition de graphes, graphes multi-échelles, choix et interprétation de graphiques, analyse, mise en forme de données.
9. Tableaux croisés dynamiques (TCD), graphiques croisés dynamiques (GCD).
10. Macros enregistrés.
11. Initiation aux macros programmés en VBA.
12. Raporting, création d'indicateurs clés, tableaux de bords, méthodologie, analyse de données, aide à la décision, études de cas.

Menus multi-déroulants de saisie

- Appelés aussi *listes déroulantes*, elles permettent à l'utilisateur de choisir une valeur parmi une sélection prédéfinie. Ça permet d'éviter les **erreurs de saisie**.
- Sélectionner une première cellule où la liste déroulante va apparaître.
- **Données → Validation des données → Autoriser : Liste.**
- Sélectionner la plage contenant les valeurs des catégories.
 - Par exemple, une plage contenant les noms des IRIS (Setosa, Versicolor, Virginica).
- Pour des sous-catégories:
 - Dans le **Name Manager**, nommer une plage de valeurs (ex: A2:A51) avec l'un des noms des IRIS (ex: Setosa).
 - Sélectionner une autre cellule, dans **Data Validation**, saisir **=INDIRECT(1^{ère} cellule)**

Verrouillage de cellules / feuilles / classeur

- Permet de *protéger* des cellules pour éviter qu'elles ne soient modifiées par les utilisateurs.
- Sélectionner les cellules à protéger.
- **Review → Allow Users to Edit Ranges.**
 - Sélectionner les plages qui doivent rester modifiables.
- **Protect Sheet.**
 - On peut choisir des actions à autoriser (formatage, suppression de colonnes, ...).
 - Saisir un mot de passe pour protéger la déverrouillage de la feuille.
- On peut appliquer une protection sur tout le classeur : **Protect Workbook.**